
1 

SESUG 2020 Paper 106 

Using Proc Report to Produce Easy to Understand Student 
Performance and Progress Analyses for Educators 

Proc Report for Student Performance Analysis 
Peter L. Zsiga, Ph D., St Lucie County Schools 

ABSTRACT  
The Florida Department of Education evaluates schools and school districts in 11 specific 
categories including student performance on statewide standardized tests. Decisions on 
financial rewards or additional reporting requirements and third-party monitoring visits are 
made based on the results. One Florida school district with over 40,000 students uses 
Enterprise Guide® software to produce detailed reports of key elements to identify students 
who may need additional support to make progress and demonstrate proficiency on the 
annual assessment. A series of if-then conversions are used before compute statements are 
applied in a Proc Report to produce an ODS Excel file with color-coded data for ease of 
interpretation. Basic and early intermediate level SAS® users should find something new 
and applicable to education and other fields.  

INTRODUCTION  
School Administrators have huge responsibilities and portfolios which include student safety, 
facilities management, personnel issues, multiple state and local regulation satisfaction, 
community involvement, parent communication, as well as coordinating student 
transportation and nutrition programs. These are all in addition to their primary 
responsibility of insuring a proper education for the students under their care. Producing 
easy to understand reports that take and make sense of a variety of student score reports 
and provide some indication of future results can be of tremendous value to them. The 
processes below apply input, output, and if-then statements; an occasional do loop; sorting, 
ranking, filtering and other processing steps; and finally Proc Report and ODS statements to 
build a color coded excel file for each school site.   

THE PROBLEM 
State accountability systems for K-12 schools measure multiple subjects across several 
grades from Third Grade to Graduation. The Florida Department of Education evaluates 
schools and school districts in 11 specific categories including student performance on 
statewide standardized tests in English Language Arts, Mathematics, Science, and Social 
Studies, and student growth in English Language Arts and Mathematics for the total tested 
population and for the students who placed in the lowest 25th percentile on the previous 
year’s assessment.  
Since six of the eleven items involve the two basic subjects, English Language Arts (ELA) 
and Mathematics, schools are proactively monitoring student performance and growth with 
periodic assessments. They need to frequently analyze the results for individual students, 
grade levels, schools, teachers, and other groupings.  
One Florida school district with over 40,000 students uses Enterprise Guide® software to 
produce detailed reports of key elements to identify students who may need additional 
support to make progress and demonstrate proficiency on the annual assessment. Both ELA 
and Mathematics reports are compiled. An inherent complication is the different scale scores 
the State applies to each subject and grade combination, and the division of these scores 
into eight separate bucket groupings as illustrated in Table 1 below. Further, the method 
used to determine whether a student has demonstrated growth since the previous 
assessment changes in the top three of the eight groupings.  

In addition, projections from local assessments differ for students who are enrolled in the 
higher-level subjects before or after entering High School. The Mathematics report is 
complicated by the fact that Grades 3 to 8 Mathematics, Algebra 1 and Geometry are all 
included, and they use a different scale to score the related assessments. This report 
discusses only the ELA report.  
Five demographic items and 11 raw scores are the inputs for ELA and produce a report with 
10 converted elements, 6 calculated items and 6 of the original raw scores and the five 
demographics. Color coding of 12 of the items allows for quick visual interpretation and 
analysis of individual and grouped student data and progress over the series of state and 
local assessments. A synopsis of the data conversion and compute statements are explored 
in the following sections.  
Reports of this type have helped elevate one Florida School District by 96 points on an 1100‐point scale 
over three years, from 573 to 669. Vastly more important than the score itself is the increase in the 
student growth and performance items each of the points represents.  
 
Table 1. Bucket Minimums by Grade Level for ELA and Math 

ELA  1.1  1.2  1.3  2.1  2.2  3  4  5 

Grade 3  240  255  270  285  293  300  315  330 

Grade 4  251  267  282  297  304  311  325  340 

Grade 5  257  273  289  304  313  321  336  352 

Grade 6   259  276  293  309  318  326  339  356 

Grade 7   267  284  301  318  326  333  346  360 

Grade 8  274  290  306  322  330  337  352  366 

Grade 9  276  294  311  328  336  343  355  370 

Grade 10  284  301  318  334  342  350  362  378 

Math  1.1  1.2  1.3  2.1  2.2  3.0  4.0  5.0 

Grade 3  240  255  270  285  291  297  311  327 

Grade 4  251  267  283  299  305  310  325  340 

Grade 5  256  273  290  306  313  320  334  350 

Grade 6   260  277  294  310  318  325  339  356 

Grade 7   269  285  301  316  323  330  346  360 

Grade 8  273  290  306  322  330  337  353  365 

Algebra 1  425  446  467  487  492  497  518  532 

Geometry  425  446  466  486   493  499  521  533 
 

DATA MANAGEMENT IS THE SOLUTION 
 

Choosing and converting selected data elements that allow for the greatest degree of 
confidence in reports to school and district administrators is a key step in data 
management. Educational software and assessment providers have over a million 
datapoints to use to produce regression equations to establish a range of predictions with a 
high degree of reliability. Those predictions can then be applied to develop school and 
district reports based on their individual student populations. Analysis of other assessments, 
including teacher or subject specialist created assessment, against a standard such as a 
related statewide test can provide another set of comparative values. 

DATA COLLECTION 
 
The input demographic data includes student name, grade, school, an identification number, 
and an indication of whether the student’s results will be attributable to the school or district 
based on their enrollment dates. The score inputs begin with two years of state test scores, 
two years of the critical writing component of those scores, and five sequential reading 
diagnostic assessment scores, and end with two years of averaged scores on the local 
assessments aligned to the district curricula. Figures 1 and 2 below show the truncated 
input file with over 30,000 records and then the results of the varnum file input statement 
to list the variables in order of creation. 
 

 
 
Figure 1. Truncated Input File 
 

 
 
Figure 2. Input Variables in Creation Order 
 
The grade is the only demographic variable in a number format along with all the score 
inputs. Variables ‘ELAXXXX’ and ‘WritingXXXX’ are the past two years of the state end of 
year ELA assessment and the writing component of that test, which has an inordinate 
influence on the students’ placement. The five variables beginning with ‘diagOSSe’ are 
periodic diagnostic evaluations of the students’ reading levels. The final variables, 
‘PredProfXX,’ reflect the mean of all the locally developed assessments for the school year 
indicated. 
 
DATA CONVERSIONS 

While the first four score inputs (#6 to 9) are displayed in the final report as they are 
entered, every other element is processed in either a data step or a compute statement in 
Proc Report, or both.  
The five diagnostic scores all use identical scales applied to if-then processes to convert 
them to one of the 8 buckets displayed above. A similar process is applied to reformat the 
local assessment averages to the buckets. All the conversion input codes are adjusted to 
reflect the scale applied to the student’s grade in the year that the score was earned. As an 
example, a current 6th grade student has the 4th grade scale applied to their 2018 score, the 
5th grade score to 2019, and their current grade to the 2020 predictions. 
The abbreviated code to convert the diagnostics to the buckets shown in Table 1 above is 
below. The diagOSSeF19 is the score from the Fall 2019 assessment and the IRBeF19 is the 
conversion to one of the eight buckets.  
 

data IRB1920;  
set ELAIRFSA1920; 

IRBeF19 = .; 
if diagOSSeF19 eq . then IRBeF19 = .;  
else if (grade eq 3 and diagOSSeF19 ge 598) then IRBeF19 = 5;  
else if (grade eq 3 and diagOSSeF19 ge 564) then IRBeF19 = 4;  
   *Continues by grade and score input with bucket output* 
else if (grade eq 9 and diagOSSeF19 ge 484 ) then IRBeF19 = 1.2;  
else if (grade eq 9 and diagOSSeF19 le 483 and diagOSSeF19 ne . ) then 
       IRBeF19 = 1.1; 
else if diagOSSeF19 eq . then IRBeF19 = . ;  
else IRBeF19 = .;  

 
The most recent diagnostic, DiagOSSeW, undergoes one additional conversion using decile 
scale to predict the probability of success on the next state assessment. The conversion 
scale factors are from an analysis by the provider of the diagnostic software. A sample of 
that conversion code is exhibited here, where IRBwePR is the probability of success:  
 

IRBwePr = .;  
if DiagOSSeW eq . then IRBwePR = .;  
if (grade eq 3 and DiagOSSeW ge 550) then IRBwePr = 90 ; 
else if (grade eq 3 and DiagOSSeW ge 537) then IRBwePr = 80 ; 
else if (grade eq 3 and DiagOSSeW ge 528) then IRBwePr = 70 ; 

  *Continues by grade and score input with decile output* 
else if (grade eq 8 and DiagOSSeW ge 554) then IRBwePr = 10 ; 
else if (grade eq 8 and DiagOSSeW lt 554 and DiagOSSeW ne .) then 
      IRBwePr = 0 ;  
else if DiagOSSeW eq . then IRBwePr = . ;  
else IRBwePr = .; 

run; 
 
Converting the two prior years test scores into the eight buckets involves a similar process 
and code as the diagnostic conversions with a different scale applied. Input variables 
ELA2018 and ELA2019 are the two years of scores used to produce the output variables 
EAL19 and EAL18 which are the eight buckets for those years. The conversion points are 
adjusted to match the students’ grade levels in 2018 and 2019.  
Those two years of test scores are then used to identify a minimum target score to 
demonstrate growth. The wrinkle here is that while students earn growth points by moving 
up a bucket, students in the highest three buckets can also show growth by remaining in 
the same bucket in the next grade and advancing one point in their score.  This requires 
adding an additional discriminant in several lines to calculate the prior year score plus one 

as the minimum for the if-then statement. As shown below, the target score can either be a 
specific number or calculated value. 
 
   ELGtarget19 = . ;  
  if grade = 3 and ELA2018 ge 330 then ELGtarget19 = 330 ;  

else if grade = 3 and ELA2018 ge 315 then ELGtarget19 = ELA2018 + 1 ;  
else if grade = 3 and ELA2018 ge 300 then ELGtarget19 = ELA2018 + 1 ;  
else if grade = 3 and ELA2018 ge 293 then ELGtarget19 = 300 ;  

  *Continues by grade and score input with target score output* 
else if grade = 10 and ELA2018 gt 0 and ELA2018 lt 294 then ELGtarget19  

= 301 ;  
else ELGtarget19 = . ;  

 
According to the State’s guide to calculating school grades, a Learning Gain is achieved if 
the attained score meets or exceeds the target score, and Proficiency is measured by 
placing in bucket 3 or higher. A Proficiency Drop indicates the student was proficient in 2018 
and had an unsatisfactory score last year. These are some of the simplest code segments, 
producing different results if the scores rose, dropped or were missing:  
 
   LG19 = ' ';  

if ela2018 eq . or ela2019 eq . then LG19 = ' ' ;  
else if ELGtarget19 gt ela2019 then LG19 = 'N' ; 
else if ela2019 ge ELGtarget19 then LG19 = 'Y' ; 
else LG19 = . ; 

   L345 = " "; 
If EAL19 ge 3 then L345 = "Y" ; 
else if EAL19 eq . then L345 = " " ; 
else if EAL19 le 2.2 then L345 = "N" ; 

 
   ProfDrop = " "; 

if (EAL18 ge 3 and EAL19 lt 3)then ProfDrop = 'Y'; 
else ProfDrop = " "; 

 
Additionally, the lowest twenty-five percent of students from each school and tested grade 
are identified for their Learning Gain separately as one of the eleven school grade 
calculation cells. Finding this bottom quartile of students requires combining retained grade 
3 and current grade 4 students into one group since their placement is based on their 
scores on the same test. A new variable, testgr, is created to combine those two grades. 
This new variable is combined with the school variable and a second variable schgrade is 
created. This process concludes with filtering for the accountability variable so only students 
who will be counted in the school grade calculation are included in the lowest quartile. For 
brevity some data steps are removed from the display of the variable creation and ranking 
code below. These steps conclude with the creation of the output variable BQELA, which 
displays a ‘Y’ if the student is ranked in the lowest 25th percentile and is also on the 
accountability list. 
 

testgr = .; 
if grade le 4 then testgr = 3; 
else testgr = grade - 1; 

   schgrade = school||testgr; 
run; 
 
   proc sort data=erank; 

by schgrade; 
run; 
 
   proc rank data=erank out=rankgrade ties=low descending; 

by schgrade; 
var ELA2019; 
ranks ELARank ; 

run; 
 
   proc rank data=rankgrade percent out=graderank ties=low descending; 

by schgrade; 
var ELARank ; 
ranks RankELA ; 

run; 
 
   BQELA = ' '; 

if (accountability eq 'Y' and ELA2019 ne . and rankela le 25 and  
rankela ne . ) then BQELA = 'Y'; 

else BQELA = ' '; 
run; 
 
An equi-percentile comparison of the 2019 year-end Predicted Proficiency from the local 
assessments and the achieved bucket for matched students produced a series of cut scores 
for each subject and grade level. Based on those results the ‘PredProf20’ variable was 
converted into one of the eight buckets to enable an easy comparison. A truncated section 
of the if-then statement code is displayed:  
 

PredProf20bkt = . ;  
  if PredProf20 = . then PredProf20bkt = . ; 

else if ( grade = 3 and PredProf20 ge 86 ) then PredProf20bkt = 5 ; 
else if ( grade = 3 and PredProf20 ge 75 ) then PredProf20bkt = 4 ; 

*Continues by grade and score input with PredProf20bkt output*  
else if ( grade = 10 and PredProf20 ge 34 ) then PredProf20bkt = 1.2 ; 
else if ( grade = 10 and PredProf20 lt 34 ) then PredProf20bkt = 1.1 ; 

else PredProf20bkt = . ; 
 
Three of the data files are then merged, and a keep statement applied to reduce the output 
while maintaining most of the original and all of the created variables in preparation for the 
Proc Report step.  
 
PROC REPORT 
 
A macro is created to make it easy to change components of the Proc Report for each 
subject and to allow for quick adjustments to summary files by school, grade and district. 
The macro sets the parameters of the Proc Report so it can be invoked for all students or for 
each school separately.  
The listing of columns in the order they are to be displayed appears first, then the Define 
statements which alter the column header label and the column width as appropriate. A snip 
of these code segments are below:  
 
***Creating the Macro*** 

%MACRO IRFSAbkt2020 ;  
  
column student School Grade FLEID Accountability ELA2018 ELA2019 
Writing2018 Writing2019 EAL18 IRBeF19 IRBeW19 IRBeS19 PredProf19 EAL19 
IRBeF IRBwe DiagOSSeW IRBwePr PredProf20 PredProf20bkt LG19 ProfDrop L345 
ELAbq ELGtarget20 ELAptsNeeded ;  
  
Define student / display "Student" ;  
Define School / display "School" ;  

Define Grade / display "grade" style(column)={width=10};  
Define FLEID / display "FLEID" style(column)={width=105};  
Define Accountability / display "Accountability" style(column)={width=30 };  
Define ELA2018 / display "FSA ELA Spring 17-18" style(column)={width=30} ;  
Define ELA2019 / display "FSA ELA Spring 18-19" style(column)={width=30} ;  

  *Continues with defined output for each column* 
Define LG19 / display "Gain Last Year" style(column)={width=30} ;  
Define ProfDrop / display "Proficiency Drop" style(column)={width=30};  
Define L345 / display "Proficient in 2019" style(column)={width=30} ;  
Define ELAbq / display "Low 25% ELA 2019-20" style(column)={width=30};  
Define ELGtarget20 / display "Gain Target 2020" style(column)={width=30};  
Define ELAptsNeeded / display "Point Growth Needed 2020"  

style(column)={width=30}; 
 
To simplify interpretation, color-coding the key variables allows for a quick visual reading of 
the trends for individual and groups of students. The colors are added by applying a series 
of compute statement in the Proc Report. The same eight color scheme is applied to the two 
‘EALXX’, the five ‘IRB..’, and the ‘PredProf20bkt’ variables, and a shorter three color scheme 
to the ‘PredProfXX’s:  
 
Compute EAL19;;  
if EAL19 = 1.1 then call define(_col_,"style","style=  
  {background=red}") ;  

else if EAL19 = 1.2 then call define(_col_,"style","style=  
  {background=coral}") ;  

else if EAL19 = 1.3 then call define(_col_,"style","style=  
  {background=pink}") ;  

else if EAL19 = 2.1 then call define(_col_,"style","style=  
  {background=orange}") ;  

else if EAL19 = 2.2 then call define(_col_,"style","style=  
  {background=yellow}") ;  

else if EAL19 = 3 then call define(_col_,"style","style=  
  {background=green}") ;  

else if EAL19 = 4 then call define(_col_,"style","style=  
  {background=blue}") ;  

else if EAL19 = 5 then call define(_col_,"style","style=  
  {background=violet}") ;  

endcomp;  
 

Compute PredProf19;;  
if PredProf19 = . then call define(_col_,"style","style=  
  {background=white}") ; 

else if PredProf19 >= 80 then call define(_col_,"style","style=  
  {background=green}") ;  

else if PredProf19 >= 60 then call define(_col_,"style","style=  
  {background=yellow}") ;  

else if PredProf19 >=0 then call define(_col_,"style","style=  
  {background=red}") ;  
endcomp;  
 
One of the predictive variables is in a decile form, requiring ten different colors to delineate 
the separate values, and the use of a white background for any empty cells. Ten hues of 
gradually deepening intensity were selected from the Predefined Color webpage in the SAS 
Help Center and a Compute code segment added:  

Compute IRBwePr;;  
if IRBwePr = .  then call define (_col_,"style","style={background=  

white}") ;  
else if IRBwePr= 90 then call define(_col_,"style","style={background= 

RoyalBlue}"); 
else if IRBwePr= 80 then call define(_col_,"style","style={background= 

DodgerBlue}"); 
else if IRBwePr= 70 then call define(_col_,"style","style={background= 

DeepSkyBlue}"); 
else if IRBwePr= 60 then call define(_col_,"style","style={background= 

LightSkyBlue}"); 
else if IRBwePr= 50 then call define(_col_,"style","style={background= 

SkyBlue}"); 
else if IRBwePr= 40 then call define(_col_,"style","style={background= 

LightBlue}"); 
else if IRBwePr= 30 then call define(_col_,"style","style={background= 

PowderBlue}"); 
else if IRBwePr= 20 then call define(_col_,"style","style={background= 

PaleTurquoise}"); 
else if IRBwePr= 10 then call define(_col_,"style","style={background= 

LightCyan}"); 
else if IRBwePr= 0 then call define(_col_,"style","style={background= 

Azure}"); 
endcomp;    

  
Finally, since students who were in the top halves of the score buckets on the most recent 
state assessments would be more likely to reach their target score on the next test, these 
students were identified and their ‘ELAptsNeeded’ variable had a color marker applied. Only 
students who had not achieved bucket 3 or higher proficiency, were selected for this 
marker. Once again, their grade level was a discriminant, along with their prior year bucket, 
and the midpoint of that bucket for the tested grade level. This section applied Do 
statements to section off each of the current grade levels in the Compute code segment:  
 

compute ELAptsNeeded ;; 
if grade eq 3 then do; if ELAptsNeeded = . then call define  

(_COL_,'style','style=[background=white]');  
 else if ( EAL19 eq 1.1 and ELA2019 ge 247 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  
 else if ( EAL19 eq 1.2 and ELA2019 ge 262 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  
 else if ( EAL19 eq 1.3 and ELA2019 ge 277 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  
 else if ( EAL19 eq 2.1 and ELA2019 ge 288.5 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  
 else if ( EAL19 eq 2.2 and ELA2019 ge 296 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  
 end;  

*Continues by grade, bucket and target score midpoint input * 
else if grade eq 10 then do; if ELAptsNeeded = . then call define  

(_COL_,'style','style=[background=white]');  
 else if ( EAL19 eq 1.1 and ELA2019 ge 284.5 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  
 else if ( EAL19 eq 1.2 and ELA2019 ge 302 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  
 else if ( EAL19 eq 1.3 and ELA2019 ge 319 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  
 else if ( EAL19 eq 2.1 and ELA2019 ge 331.5 ) then call define  

(_COL_,'style','style=[background=Fuchsia]');  



9 

 else if ( EAL19 eq 2.2 and ELA2019 ge 339 ) then call define  
(_COL_,'style','style=[background=Fuchsia]');  

 end;  
endcomp; 

run; 
****Ending the Macro***** 
%mend  IRFSAbkt2020  ; 
 
After the ‘IRFSAbkt2020’ macro is created, it is invoked in the Proc Report step in the ODS 
segment: 
 

ODS EXCEL    
file='C:\Users\pzs0116\OneDrive - St. Lucie Public Schools\ADAM\SESUG\SESUG 
20\IR FSA ELA BLTs 4 SESUG2020 2.xlsx' STYLE=seaside OPTIONS ( Orientation 
= 'landscape' FitToPage = 'yes' Pages_FitWidth = '1' Pages_FitHeight= '100' 
row_heights="155px" Flow='Headers' FROZEN_HEADERS='yes' gridlines='yes' ); 
 
Proc Report  data=elabq nowd style(header)={tagattr="rotate:90"};  
%IRFSAbkt2020 ;  
run;  
quit;  
ods excel close; 
 

School, grade, or other specific reports can be produced with the inclusion of a where 
statement before the macro clause in the Proc Report. 
 
RESULTS 
 
As illustrated in Figure 3, even a cursory glance at the finished report shows areas of green 
and blue which represent proficiency, and reds, yellows and oranges which indicate various 
levels of concern. The arrows point to the prior year bucket rating and the predicted rating 
for the upcoming test. The first three and last two students in view are predicted to show a 
learning gain. Student 5 is expected to maintain proficiency, while student 4 is predicted to 
drop, and may need additional support. 

 
Figure 3. Truncated Output Excel File 

Filtering and sorting features give the school administrators the options to group their 
students by grade or any of the actual or predicted performance measures. This allows for 
categories of students to be identified for instructional adaptations designed to support 
struggling students.  

CONCLUSION 
 
The process provides school administrators and teachers valuable information about their 
students and their performance relative to other students and statewide assessment results. 
Each of the conversion steps applies a generic calculation to produce an adjusted value. Any 
degree of error in the initial data or calculations will always produce some inconsistencies. 
While data can reflect trends and areas for further analysis, educators know that the most 
inconsistent variable in any report is the student. The final determination on how to react to 
the report is always in the hands of those who are closest to the students. 

